

FRP301 Advanced Excel: Instructor Notes

Table of Contents

Updated 11/2/2016

| | |
|--|----|
| 1) Introduction Review..... | 1 |
| 2) COUNTIF/SUMIF..... | 2 |
| 3) Using IF..... | 4 |
| 4) Large and Small values..... | 6 |
| 5) VLOOKUP..... | 7 |
| 6) Pivot Tables..... | 11 |
| 7) Filtering with AND and OR..... | 16 |
| 8) SUMPRODUCT..... | 19 |
| 9) Conditional Formatting – Dynamic Row Banding..... | 20 |
| 10) DataBase Functions..... | 21 |
| 11) Miscellaneous Items..... | 22 |
| 12) FRP301 Formula..... | 23 |

1) Introduction Review

a. Named Ranges

1. Refers to
 - (1) Cells
 - (2) Range of cells
 - (3) A constant value
 - (4) A formula
2. An absolute reference that can access from anywhere in a workbook
3. Replaces cell references
4. Two ways to create a range
 - (1) Manually – Highlight the range. Click into the Name Box, overwrite the cell contents with the named range (alphas, numbers, underscores only, no spaces allowed). Manually naming a range is useful for creating a table of data.

- (2) Automatically – (best used for a single column or row of data). Place your cursor in the Header field of the column, copy down (Ctl-Shift down arrow), open the dialog (Ctl-Shift F3), Click OK

2) COUNTIF/SUMIF

a. COUNTIF

- COUNTIF looks at a range of data and counts the items that equal the criteria
 - Syntax: `=COUNTIF(RangeToLookAt,Criteria)` The range can be a range of cells (e.g., A1:A40) or a Named Range from above. The criteria can be a word or number (e.g., "13010" and it must be in double quotes), A named cell, a formula (e.g., ">0"), or a cell reference that contains the criteria
 - Example: `=COUNTIF(TRC,J4)`
- Other uses
 - Count all Non Blanks: `=COUNTIF(range,"<>")`
 - Count all Text: `=COUNTIF(range,"><")`
 - Count all Text: `=COUNTIF(range,"*")`
 - Count all Blanks: `=COUNTIF(range,"")`
 - Count all Blanks: `=COUNTIF(range,"")`

b. SUMIF

- SUMIF looks at a range of data in one columns and then adds up the contents of rows in another column that is related to the first column
- Syntax: `=SUMIF(RangeToLookAt,Criteria,RangeToSum)` The range can be a range of cells (e.g., A1:A40) or a Named Range from above. The criteria can be a word or number (e.g., "13010" and it must be in double quotes), A named cell, a formula (e.g., ">0"), or a cell reference that contains the criteria. The RangeToSum contains the related row cells that will be added
- Example: `=SUMIF(TRC,J4,Quantity)`

c. Count Multiple OR Criteria

- Syntax: `=SUM(COUNTIF(Range,{criteria1,criteria2,criteria3,etc}))` Use this when you have a lot of OR criteria
- This does not need to be array entered
- `=COUNTIF(range,"criteria")+COUNTIF(range,"criteria")` will also work. This should be used when there are only a few OR criteria
- The criteria can be a word or number (e.g., "13010" and it must be in double quotes), A named cell, a formula (e.g., ">0"), or a cell reference that contains the criteria

d. Sum Multiple OR Criteria

- Syntax: `=SUM(SUMIF(Range,{criteria1,criteria2,criteria3,etc},RangeToSum))`

2. This does not need to be array entered
3. `=SUMIF(range,criteria,RangeToSum)+SUMIF(range,criteria,RangeToSum)` will also work. This should be used when there are only a few OR criteria
4. The criteria can be a word or number (e.g., "13010" and it must be in double quotes), A named cell, a formula (e.g., ">0"), or a cell reference that contains the criteria.

e. COUNTIFS (Excel 2007 and higher)

1. COUNTIFS looks at multiple related ranges and provides a count of the items that meet the related criteria. The range and criteria arguments are paired in the function.
2. Syntax: `=COUNTIFS(Range1,Criteria1,Range2,Criteria2,...)`
3. Example: `=COUNTIFS(DAY,K$23,TRC,$J24)`

f. SUMIFS

1. SUMIFS, similar to COUNTIFS, looks at multiple related ranges and criteria and provides a sum of the cell rows in a related range. SUMIFS is different from SUMIF in that the RangeToSum argument appears first and is required.
2. Syntax: `=SUMIFS(RangeToSum,Range1,Criteria1,Range2,Criteria2,...)`
3. Example: `=SUMIFS(Quantity,DAY,K$36,TRC,$J37)`

g. SUMPRODUCT

1. The SUMPRODUCT function works as both SUMIFS and COUNTIFS depending on how it is configured. SUMPRODUCT is based on the array concept but doesn't require array entry in order to work. Unlike SUMIFS or COUNTIFS, it is Range = Criteria and does not use the standard comma argument separator.
2. Syntax:
 - (1) `=SUMPRODUCT((Range=Criteria)*(Range=Criteria))` if used to count items,
 - (2) `=SUMPRODUCT((Range=Criteria)*(Range=Criteria)*RangeToSum)` if used to sum items.
3. Examples
 - (1) `=SUMPRODUCT((DAY=K$23)*(TRC=$J24))` ~ used as a count
 - (2) `=SUMPRODUCT((DAY=K$36)*(TRC=$J37)*Quantity)` ~ used to sum
4. See also Section 6

3) Using IF

a. IF provides for an action depending upon whether a test is true or false. In general, the syntax is “If the test is true, then perform the action for true, and if the test is false then perform the action for false.” For example, in commissioned sales commissions are based on the amount of sales made. If a salesperson sells over \$10,000 in a period then they get 7% and if they sell less, they get 5%. The test is, are sales greater than \$10,000? If the test is true then multiply the amount of sales by .07 (7%) to arrive at the commission. If the test is false, then multiple the amount of sales by .05 (5%) to arrive at the commission.

b. IF

1. Syntax: `=IF(ConditionToTest,ActionIfTrue,ActionIfFalse)`. The actions can be a formula, a word or phrase (in quotes, “True”), or the contents of a cell.
2. AND/OR
 - (1) AND tests multiple criteria and if they are all true then it returns TRUE as a result. If any part of the test is false then it returns FALSE. Or test multiple criteria and if any individual test is true then it returns TRUE. If all of the tests are false then it returns FALSE.
3. IF-AND allows for multiple tests. What needs to be remembered is that if any test is false then it is all false.
 - (1) Syntax: `=IF(AND(Test1,Test2,...),ActionIfTrue,ActionIfFalse)`. Note that there are multiple parenthetical phrase (one for IF and one for AND). Any time a parenthesis is opened it must be closed. You cannot have an odd number open versus closed parentheses.
4. IF-OR is similar to IF-AND in that it allows for multiple test conditions. The difference is the with the OR function if any test it True then everything is True.
 - (1) Syntax: `=IF(OR(Test1,Test2,...),ActionIfTrue,ActionIfFalse)`

c. IF-ISERROR / IFERROR

1. ISERROR is something of opposite function. It tests a formula and if the result of the test is an error (e.g., $0 \div 0$) then it returns TRUE. This means that when combined with IF, you have to think of the ActionIfTrue as what you want to happen if the test results in a error.
2. Syntax: `=IF(ISERROR(ConditionToTest),ActionIfTrue,ActionIfFalse)`
3. IFERROR was developed for Excel 2007/2010 to accomplish the IF-ISERROR function in one function instead of using two.
 - (1) Syntax: `=IFERROR(ConditionToTest,ActionIfFalse)`. If the ConditionToTest is true the function will insert the results of the test.

e. IF Nested

1. If Nested allows for multiple IF tests. If a test is true then the ActionIfTrue is invoked. If the test is false then the ActionIfFalse is another IF. In Excel 2003 or lower you can have up to 7 If nestings. For Excel 2007/2010 you can have up to 64. Since you using multiple IFs there are multiple parentheses which need to be opened and closed. There have to be an equal number of open and closed parentheses.

2. Syntax:

```
=IF(Test1,ActionIfTrue,IF(Test2,ActionIfTrue,IF(Test3,ActionIfTrue,ActionIfFalse)))
```

3. Formula from the worksheet:

```
=IF(E3="10010",O3*1.05,IF(E3="10020",O3*0.9,IF(E3="10050",O3*0.5," "))),
```

```
=IF(E3=payroll,O3*pay1,IF(E3=oe,O3*oe1,IF(E3=equip,O3*equip1," "))) – uses  
named cells,
```

```
=O3*VLOOKUP(E3,table,2,0) – uses a VLOOKUP
```

```
=IF(ISERROR(VLOOKUP(E3,table,2,0)),"",O3*VLOOKUP(E3,table,2,0))
```

```
=IFERROR(O3*VLOOKUP(E3,table,2,0),"")
```

f. IF Other Uses

1. To find the Nth occurrence of an item. The important thing to note is that you have to anchor one end of the range (doesn't matter which end) and leave the other end to cascade. The syntax is: =IF(COUNTIF(range,criteria)=N,criteria,"") where the range is anchored at one end and cascades at the other (i.e., E2:\$E\$2)

(1) 1st instance =IF(COUNTIF(\$E\$2:E2,E2)=1,E2,"")

(a) or =VLOOKUP(A2,\$E\$2:\$F\$11,2,0) in B2

(2) 2nd instance =IF(COUNTIF(\$E\$2:E2,E2)=2,E2,"")

(3) 3rd instance =IF(COUNTIF(\$E\$2:E2,E2)=3,E2,"")

(4) Last instance =IF(COUNTIF(\$E\$2:E2,E2)=COUNTIF(\$E\$2:\$E\$28,E2),E2,"")

The part in red is the condition to be tested. Note that the range in the first half of the condition is semi-locked, and the range in the second half of the condition is fully locked.

4) Large and Small values

a. Concept

1. This section deals with getting information from a list based on the values that are in the list.

b. RANK

1. Syntax: =RANK(CellRef,Range,1 or 0) where 1 = ascending order and 0 = descending order
2. Example: =RANK(A7,List,1) returns 3 as the list ranking based on ascending order
3. Example: =RANK(A7,List,0) returns 29 as the list ranking based on descending order.

c. MIN, MAX

1. Syntax
 - (1) =MAX(list) returns the largest value in the list: 496
 - (2) =MIN(list) returns the smallest value in the list: 100

d. LARGE, SMALL

1. Syntax:
 - (1) =LARGE(list,RankPlaceSought)
 - (a) Example: =LARGE(list,2) returns the second largest value: 487
 - (2) =SMALL(list,RankPlaceSought)
 - (a) Example: =SMALL(list,2) returns the second smallest value: 117

e. Formulae

1. Sort a list largest to smallest: =LARGE(List,ROWS(\$A\$7:A7))
 - (1) You can also use =LARGE(List,ROW(A1))
2. Sort a list smallest to largest: =SMALL(List,ROWS(\$A\$7:A7))
 - (1) You can also use =SMALL(List,ROW(A1))
3. Sum the top ten values: =SUMIF(List,">="&LARGE(List,10))
4. Sum the top three values: =SUM(LARGE(A7:A37,{1,2,3}))
 - (1) =SUM(LARGE(List,{2,4,6})) sums the second, fourth, and sixth largest values
 - (2) Substitute SMALL for small values
 - (3) =SUM(LARGE(List,{1;2;3;4;5}))+SUM(SMALL(List,{1;2;3;4;5})) sums the top five plus the bottom five values.
5. Sum the even numbers in a list: =SUMPRODUCT(--(MOD(List,2)=0),List)
 - (1) =SUMPRODUCT(--(MOD(List,2)=1),List) sums the odd numbers in a list
6. Count the even numbers in a list: =SUMPRODUCT(--(MOD(List,2)=0))
 - (1) =SUMPRODUCT(--(MOD(List,2)=1)) counts the odd numbers in a list
7. Array Formulae (Ctrl-Shift-Enter)
 - (1) Sum the odd rows: {=SUM(IF(MOD(ROW(List),2)=1,List,0))}
 - (2) Sum the even rows: {=SUM(IF(MOD(ROW(List),2)=0,List,0))}

(3) Display the top five values when a range is highlighted first:

{=LARGE(List,{1;2;3;4;5})} (Note the semi-colons)

8. Conditional Formatting (highlight the list)

(1) =A7>=LARGE(LIST,5) shows top five

(2) =A7<=SMALL(LIST,5) shows bottom five in the same list

5) VLOOKUP

a. **VLOOKUP** is made up of four arguments: Criteria, Lookup table, column to return, TRUE or FALSE

b. Syntax: =VLOOKUP(Criteria,LookupTable,ColumnToReturn,TRUEorFALSE)

1. Criteria is the item you want to look up. This can be a word or number (in quotes, “word”), or a cell reference.

2. The lookup table is where the answer is located and is generally two or more columns, and the criteria (see above) must be in the left-most column. The lookup table can be a range of cells or a named range. A named range is preferred because it provides a locked block of cells.

3. The column to return contains the information you want to return that is related to the criteria and will be on the same row as the criteria.

4. TRUE or FALSE

(1) FALSE tells the function to find an exact value in the lookup table. If it cannot find an exact match then it returns #N/A. the lookup table does not need to be in any particular sort order.

(2) TRUE tells the function to find an exact match and if there is no exact match then return the next lowest item in the table. The lookup table must in an ascending sort order. For example, you have a lookup table where 97 is associated with A+, and your criteria is 98. 98 is not in the lookup table but 97 is, so the function will return the A+ as the found value.

c. Special Note ~ If you have text in one place and numbers in the other place then the VLOOKUP will fail. The below presents several ways to make the data the same without converting the tables themselves. The first two work best when the lookup is a text and the table is a number (e.g., 000123 vs 123). They will fail if the values don't “look” the same (e.g., 123 (as a number) vs 123 (as a text))

1. =VLOOKUP(VALUE(D2), A2:B9,2,FALSE)

(1) VALUE converts a text string to a number (e.g., 123 formatted as text)

2. =VLOOKUP(0+(D2), A2:B9,2,FALSE) or

=VLOOKUP(1*(D2), A2:B9,2,FALSE)

=VLOOKUP(--LEFT(D2,5),A2:B9,2,FALSE)

- (1) The first argument forces a text string to become a number string for search purposes.
- (2) Assumes that the lookup table has a numerical column 1
- 3. =VLOOKUP(TEXT(A2,"000000"),TEXT(table,"000000"),column,FALSE) forces both the lookup and the table to be text.
 - (1) Requires array entry: Ctrl+Shift+Enter
 - (2) This formulation is recommended if your data is mixed. For example, your lookup is 123 but your table value is 000123
 - (3) Warning. This will have an odd effect if the return value is another number.
 - (4) =INDEX(rng_1,MATCH(TEXT(D2,"00000000"),INDEX(TEXT(rng_1,"00000000"),,1),0),2) solves the text problem, but has its own disadvantages.
 - (5) Ultimately, where the lookup value is a number stored as text is fairly easy to solve by using any answer in 1 or 2 above. Where the lookup table has a number stored as text is more difficult and may require compromise on the user's part, although (4) is a good answer.
- d. Using VLOOKUP to look left ~ VLOOKUP was designed to have the lookup value in the left-most column and the return value to the right. It is possible to have the lookup value to the right and the return value to the left.

1. Syntax:

=VLOOKUP(lookup_val,CHOOSE({2,1}, result_column,lookup_column),2,FALSE) or
 =VLOOKUP(lookup_val,CHOOSE({1,2},lookup_column,result_column),2,FALSE)

- (1) In place of the lookup table you define the two columns where the return value is located and then the column where the lookup is. =CHOOSE takes an index number as its primary argument which searches for the corresponding values in the list. For VLOOKUP, the CHOOSE argument has been modified to look at two columns. The curly braces allows CHOOSE to have more than one index number, and they determine the order of how columns are chosen.

| | A | B | C | I | J | K | L | M | N | O |
|---|--------|-----------|---------------------|---|-----------------|--------|---|---|---|--|
| 1 | | | | | | | | | | |
| 2 | ID | Empl Rcd# | Name | | Name | ID | | | | |
| 3 | 589766 | 0 | Passeri, Sabrina | | Lisitano, Donna | 453393 | | | | =VLOOKUP(J3,CHOOSE({1,2},Name,ID),2,FALSE) |
| 4 | 377885 | 0 | Venditto, Kelli-Ann | | Quinn, Afreen | 042363 | | | | =VLOOKUP(J4,CHOOSE({2,1},ID,Name),2,FALSE) |

- (2) In this example, the answer to be returned is in column A, and the lookup is in column C.
- (3) =VLOOKUP(J3,CHOOSE({1,2},Name,ID),2,FALSE)
 =VLOOKUP(J4,CHOOSE({2,1},ID,Name),2,FALSE)
 These both return the same answer

2. INDEX and MATCH can do the same job.

| | A | B | C | I | J | K | L | M | N |
|---|--------|-----------|--------------|---|-------------|--------|---|----------------------------------|---|
| 2 | ID | Empl Rcd# | Name | | Name | ID | | | |
| 5 | 136307 | 0 | Rheaume,Alyn | | Kata,Camila | 131923 | | =INDEX(table,MATCH(J5,Name,0),1) | |

3. Syntax

=INDEX(lookup_table,MATCH(Row_lookup,range,type_of_match), Column_number)

(1) Type of Match:

1 = the largest value that is less than or equal to the lookup value

0 = an exact match

-1 = the smallest value that is greater than or equal to the lookup

4. Examples: (based on the graphic)

=INDEX(table,MATCH(J5,Name,0),1)

The Name range is Column C and the return value is 1, or Column A.

=INDEX(tblLookup,MATCH(R11,Name,0),MATCH(\$A\$3,column,0))

- e. In the exercise, you can write one formula without modification to cover all three columns. This will work only because the table being filled out and the source table are set up the same way.

1. In C4 enter, =VLOOKUP(\$B4,ct_tbl,COLUMN(B1),1)

(1) \$B4 locks the column but not the row, so the row will increment as it is copied.

(2) COLUMN(B1) is just a geeky way of saying "2". This will also increment when copied.

(3) Combined, the two items form a flexible formula.

2. Alternatively, you can use a helper row since the table and the sheet to be filled in are in the same order.

(1) On the tab where the answer is to appear on the row above the column headers, enter the numbers 2, 3, 4, 5

(2) The VLOOKUP is then written to lock the date column (\$B4) and the number row (C\$2): =VLOOKUP(\$B4,ct_tbl,C\$2,TRUE)

(3) If you don't want the numbers to be displayed, then highlight, press Ctl-1 (format cells), Custom Format, ;;; (this will force the numbers to be the same shade as the background ~ essentially hiding them).

f. Using VLOOKUP to pull results from another workbook





1. Required: both the receiving and the source workbooks are open

(1) To make your lives easier, make sure the source workbook has a named range as its source table

2. On the receiving workbook start the formula: =VLOOKUP(H3,

(1) Once you type the first comma, press Ctl-Tab to move to the other workbook

(2) Press F3 to open the Paste Name dialog, and select the source table. Click OK or press Enter

- (3) Type the column number, True or False, close the parenthetical, Press Enter
- (4) The result will look something like: =VLOOKUP(H3,email20140328.xlsx!table,6,0)
- The second argument is **email20140328.xlsx!table** which is the name of the workbook “!” name of source table (don’t ever try to type this)
 - Bonus: the fourth argument above is 0. Remember, in Excel FALSE = 0 so you can substitute 0 for FALSE
- g. Troubleshooting VLOOKUP errors
- VLOOKUP requires an exact match in order to return an answer. If the lookup isn’t exact, it will return #N/A.
 - VLOOKUP does not distinguish between upper and lower case. TC101 is the same as tc101. When dealing with text entries, the most common problem is spaces.
 -  is not the same as . Note the location of the cursor. In the first example, the cursor shows two extra spaces. The second example shows no spaces.
 - If the text with spaces is the lookup value then =VLOOKUP(TRIM(A9),list,2,0)
 - The TRIM function removes all spaces front and back, and will delete all but one space between words.
 - If the text with spaces is in the lookup table then =VLOOKUP("*"&A10&"*",list,2,0) or =INDEX(list,MATCH(A11,D7:D19,1),2). MATCH is more forgiving of spaces than is VLOOKUP, but you have use the inexact match so make sure your lookup table is sorted.
 - The “*” tells the VLOOKUP to look for spaces around the value.
 - You have to be careful when dealing with numbers.
 -  is not the same as 
 - The number is left justified in the first example, and right justified in the second. The left justification is an indicator that what you’re seeing is a number stored as text. The green triangle is also an indicator of a number stored as text, but you may not always have a green triangle.
 - There are two ways to correct for the number versus text problem
 - If the number as text is the lookup then
 - =VLOOKUP(0+A5,list1,2,0)
 - =VLOOKUP(1*A6,list1,2,0)
 - =VLOOKUP(--A7,list1,2,0)
 - The 0+A5, 1*A6, and –A7 all force Excel to look at the text as a number
 - If the lookup is a number and the lookup table is number as text, then
 - =VLOOKUP(TEXT(H5,"0"),list2,2,0)
 - This forces Excel to look at the lookup value as a text.

6) Pivot Tables

a. Setup

1. Navigation: **Insert Ribbon > Pivot Table: Click OK**
 - (1) Or, Alt-N-V-Enter ~ In Excel 2013 or later this will create a pivot table. In Excel 2010, it would be Alt-N-V-Enter-Enter
 - (2) Or, Alt-D-P ~ this will open the pivot table wizard. With the wizard open, just click Finish.
2. Pivot Tables have two associated tabs: Options and Design
3. Field manipulation should occur within the Pivot Table Field List area
 - (1) **SID = Column**
 - (2) **Vendor = Row**
 - (3) **Voucher Amount = Data**
 - (4) **Fund = Report Filter**

b. Filters:

1. Each item on the Pivot has a dropdown
 - (1) Report Filter
 - (2) Column
 - (a) Ctrl – to manually hide column header
 - (3) Rows
 - (a) Ctrl – to manually hide row header or detail
 - (4) Insert Slicer
 - (a) Available in Excel 2010 or higher
 - (b) Allows the user to filter data without the need to open the drop down lists
 - (c) Multiple filters are cumulative
 - (i) What's filtered out from one slicer is not available in the next

c. Columns and Row Headers

1. Click into the Column Label
 - (1) Rename to SID

d. Summary

1. By hovering your mouse over a data item there will be displayed
 - (1) The Sum of the field
 - (2) Row Header
 - (3) Column Header
2. Drilldown: Drill down to show the detail of the item on a separate worksheet
 - (1) Double Click Grainger
 - (a) Or Right click > Show Details
 - (b) The Show Detail dialog displays

e. Position

1. When you add a second field to a Field List box they display in order from the top down. If you want to change how the display looks, all you have to do move (up or down) the field in question.

f. Grouping-Data

1. This works best with dates, but other data can also be grouped. See 04-Pivot Table 3 for more information about grouping non-date data.
2. Select and right-click any date field.
3. From the menu displayed select Group
4. From the dialog box select Month and click OK
 - (1) Select days and Number of Days (set to 7) to show weeks
 - (2) Starting is usually the first date in the data. This can be set to first day of a week in order to display whole weeks
5. Quarters are Calendar quarters. If you need fiscal quarters, use the formula in the VLOOKUP exercise and a helper column

g. Grouping-Ad hoc (non-date data)

1. Non-date data can be grouped.
 - (1) Highlight the data you want to group. Right click > Group. A generically named group will appear (e.g., Group 1). This group can be rename to something meaningful.

h. Add a Field in the data table / Refresh the Pivot Table

1. Sometime you need to update the source data for a pivot. When this happens it is important to insert the new column inside the existing table. If you add a column to the immediate right of the existing table then it is outside the area defined as a table and you will need to redefine your table. If you insert your new column inside the existing data then Excel will automatically expand its table definition.
2. When you insert a new column it will pick up its formatting from the column to the left. If that column is formatted as Text then you will have to reformat the column as General.
 - (1) Highlight the column > Ctl-1 (opens the format cells dialog) > select general > click OK
 - (2) In the example, enter the formula: PO Amount – Voucher Amount
 - (3) Give the column Header a name. The Pivot table cannot update if there is no column name. (in the example: Remainder 1)
3. Refresh the Pivot table
 - (1) **Navigation: Pivot Table Tools > Options > Refresh**
 - (2) The new field will display in the Field List and can be used for pivot summary purposes

i. Hiding Zero Values

1. When you only want to see numeric fields that have a value
 - (1) Check the box so it appears in the Values area
 - (2) Drag and drop the field to the Report Filter
 - (3) Report Filter drop-down
 - (a) Check the Select Multiple Items
 - (b) Uncheck "0" or "-" value
 - (c) Click OK

j. Calculated Fields

1. Calculated Fields exist only as part of the pivot. They can be considered to be a virtual field. It does not have a place in the source data
2. **Navigation: Pivot Table Tools > Options > Fields, Items, & Sets (dropdown) > Calculated Field**
3. When the dialog box opens, it will ask for the name of the field (in the example: Remainder2)
4. The second box asks for a formula (all formulae begin with =). In this example: = 'PO Amount' - 'Voucher Amount'. Click OK
5. The Calculated Field is automatically activated and made part of your pivot table. In our example, if both remainders are shown, they will display the same amounts since they were developed using the same fields. The difference is that the first actually adds a column to the source data and the second acts in a virtual capacity

k. Formatting

1. How a Pivot table is presented can be as important as the information it contains
2. Excel 2010 provides dozens of preformatted templates to overlay on your data to make it look more interesting than the standard display. I think they are all repetitious and derivative and I don't like any of them. The earlier templates from Excel 2003 were better although there were only a limited number.
3. **Navigation: PivotTable Tools > Design > PivotTable Styles.**
 - (1) You may also want to experiment with the layout area and the PivotTable Style Options.

l. Show Pages



1. Show Pages is related to the Report Filter.
2. The Report Filter can use any field that isn't already being used in the pivot. If you want to have it usable both as a report filter and as a pivot table option then you will need to add it as a field in the source data so that it appears twice.
3. Show Pages allows the user to create a separately tabbed report for each item in the selected Report Filter.

4. Navigation: PivotTable Tools > Options > Options (dropdown) > Show Report Filter Pages. A dialog box opens that allows to select which report filter you want to use, if you have more than one. Click OK
 - (1) A new tab and report is created for each item in the selected Page.

m. Page Breaks (Not Taught)

1. Set your Rows to Repeat
 - (1) Page Layout (Ribbon) > Print Titles > Rows to Repeat at Top
 - (2) Set your Page Width from Automatic to 1 Page
2. Primary Sort
 - (1) Select the field item where you want the page breaks
 - (2) Right click and select Field Settings from the menu
 - (3) Select the Layout & Print tab
 - (4) Check the box labeled Insert page break after each item
3. Secondary Page Breaks
 - (1) Select the field item where you want to place the secondary page break
 - (2) Right click and select Field Settings from the menu
 - (3) Select the Layout & Print tab
 - (4) Check the boxes: Insert blank line after each item label, and Insert page break after each item.

n. Expand Collapse Field (Not Taught)

1. If you have two or more items displayed in a pivot table, you may not want to see the second item. You may just want to see the total of what that item represents.
 - (1) To collapse a single heading: select the heading > Right Click > Expand/Collapse > Collapse
 - (2) To collapse all the headings so only the totals display: Select the Heading > Right Click > Expand/Collapse > Collapse Entire Field
 - (3) To expand a single heading: Select the Heading > Right Click > Expand/Collapse > Expand
 - (4) To expand all the headings so totals and subtotals display: Select a heading > Right Click > Expand/Collapse > Expand Entire Field
 - (5) Expand and Collapse Entire Field can also be found on the Ribbon: PivotTable Tools > Analyze > Active Field (Group) >  or 

o. Charting (Pivot Table 2) and Slicers

1. Not all data lends itself to being charted. If the data lends itself to charting, it can be very effective in telling its story.

p. Pivot Table 2

- (1) Quantity = Data Values
- (2) Day = Row

(3) TRC = Column

(4) Rpt Dt = Row (Group to 14 days starting 12/10/2010) ~ move to the Report Filter

(5) Rename Row Labels to Day

(6) Rename Column Labels to TRC

2. Highlight and copy the pivot table (A1:G10)
 - (1) Paste to J1 and A20
 - (2) Pivot A1: Pivot Tools > Options > Pivot Chart > Column
 - (a) Title: Hours. Pivot Chart Tools > Layout > Chart Title > Overlay
 - (b) Remove Chart Filters. Pivot Chart Tools > Analyze > Field Buttons > Hide All
 - (3) Pivot J1: Create a Column Chart.
 - (a) Right click the data area > Summarize Values by > Count
 - (b) Pivot Tools > Options > Pivot Chart > Column
 - (c) Title: Occurrences. Pivot Chart Tools > Layout > Chart Title > Overlay
 - (d) Remove Chart Filters. Pivot Chart Tools > Analyze > Field Buttons > Hide All
 - (4) Pivot A20: Remove TRC, Day, and Sum of Quantity.
 - (a) Pivot Table Tools > Options > Slicer > Rpt Dt > OK.
 - (b) Right click the slicer and select PivotTable Connections > Check all the boxes.
This allows the slicer to control both charts.
 - (c) Test for results
3. **Optional: Navigation: PivotTable Tools > Options > PivotChart.** The Insert Chart dialog displays. Select the Column Chart. Click OK
 - (1) Right click on the column representing Sick Time > Change Series Chart Type > Select the first Line Chart. A line chart replaces the Sick column. This chart can be enlarged by dragging any corner to resize.
 - (2) Right Click the 'Sick' line and select Data Labels. The number of hours associated with each data point is displayed.
 - (3) Don't be afraid to experiment to achieve the results you are looking for.

q. Ad Hoc Grouping: Pivot Table 3

1. Along with dates, any pivot information can be grouped, Number data can be grouped in the same manner as date data. Select a field, right click, select group. Alphabetic or alpha-numeric data can be grouped manually
2. Create a pivot table. Insert (ribbon) > Pivot Table > OK
 - (1) Sum Total Amt = Data (or Values)
 - (2) Acct = Row Labels
3. Select Accts 50710 through 50800.
 - (1) Right Click > Group. (or: Pivottable Tools > Analyze > Group Field)
 - (a) Group1 will be created. Group1 is a default name and can be changed. Click into the field and rename Group1 to Personal Svc.

- (b) Repeat for the following
 - (i) 51111 – 52542 : Sundry
 - (ii) 53011 – 53920 : Current Expense
 - (iii) 54020 – 54850 : Commodities
 - (iv) 55630 – 55730 : Assets
- (c) Notice that when you grouped the first field that Acct2 was added to the Pivot Table field list
- (2) Uncheck Acct and Acct2. Select DeptID (Manual Grouping)
 - (a) Group the following (DeptID2)
 - (i) SDE64060 - SDE64170 : East Region
 - (ii) SDE64201 – SDE64295 : West Region
 - (iii) SDE64301 – SDE64346 : North Region
 - (iv) SDE64405 – SDE64530 : South Region
- (3) Move DeptID2 to the Column area
- (4) Add Acct2 to the Row area
- (5) You have created an ad hoc summary of your data.

4. MIN-MAX: Pivot Table 4

- (1) A Pivot Table Field can be added to the Value area more than once. You might want to do this if you need both a SUM and a COUNT of the numbers summed. In this example, we want to show both a start date and an end date.
- (2) Create a pivot table
 - (a) Check the Course Check box
 - (b) Drag the Start Date to the Value Area.
 - (i) Right click > Summarize by > MIN. You will get a 5 digit number
 - (c) Drag the Start Date to the Value Area again
 - (i) Right click > Summarize by > MAX. You will get a 5 digit number
 - (d) Highlight the data in the Value area > Ctrl-1 > Date > Select the first Value > OK

7) Filtering with AND and OR

a. Filtering to show a subset of values

1. When working with large data tables, it is recommended that you use the “copy to another location” feature.
 - (1) Highlight your data table, and in the Name Box, give your table a name.
 - (2) The location you will copy to is a blank sheet (either in the same workbook or in a different workbook. If it is a different workbook, both workbooks must be open.)

2. Highlight and copy those fields on the header row you want to use as criteria and paste them where you want your answer to appear. Note: Your criteria only needs to include those items you want to use as criteria (the header is required and your criteria below). If you are not using the copy and paste feature then you must type the header and criteria exactly the same way as it appears in the data table. This becomes your criteria area.
 - (1) Enter your criteria below the related header field(s)
 - (2) AND/OR: Criteria on the same row constitute the AND criteria. Criteria on different rows constitute the OR criteria. = is assumed. Other operators include: <, <=, >=, >, <> (does not equal)
3. Place your cursor in the data area.
4. **Navigation: Data (ribbon) > Sort & Filter-Advanced** or Data > Filter > Advanced Filter (the Advanced Filter dialog box displays) (2003)
 - (1) Check the Copy To Another Location radio button (preferred and recommended)
 - (2) List Range: Self-selected so long as you placed your cursor in the data area. Otherwise, click the lookup and select the data area (including the headers). Click the lookup to return to the Advanced Filter dialog. This would be a good use of a named range.
 - (3) Criteria Range: Click the lookup and select the criteria area (include the header fields and the criteria fields).
 - (4) Copy To: Click the lookup. Select where you want the unique values to display.
 - √ If you select just the one cell then the filter will return the same number of columns as are displayed in the data area.
 - √ If you only want to return selected columns then you should set that up before you use the filter. Select the column headers you want returned and place them in the Copy To location. Using the Copy To lookup, select those headers. Click the lookup to return to the dialog
 - (5) Click OK.
5. Note: If you want to extract the result subset to another sheet, you will need to set up the criteria on that sheet, and create a named range for the data table. This will simplify the extract process.

b. Filtering for Unique Values

1. The Unique values feature is a subset of the filtering operation. Otherwise, it is set up in exactly the same way as any other filter.
2. **Navigation: Data ribbon > Sort & Filter-Advanced** (earlier than 2007) Data > Filter > Advanced Filter (the Advanced Filter dialog box displays)
 - (1) Check the Copy To Another Location radio button (preferred and recommended)

- (2) List Range: Click the lookup. Highlight the column of data that contains your data of interest. You must include the column header. If you created a named range, you type the name of the range of use the F3 (Function-3) to select the named range from the list.
 - (3) Criteria Range: Ignore
 - (4) Copy To: Click the lookup. Select where you want the unique values to display.
 - (5) Check the Unique Records Only checkbox.
 - (6) Click OK.
3. Note: If you are trying to find unique values from a calculated field, Excel does not like to deal with calculated functions (e.g., the concatenated fields because it only sees the formula and not the resulting answer). To solve this problem, you copy the range and, using the Paste Special feature paste "just values". This leaves the data while removing the calculation.
- (1) Also, this process assumes there will be a header row, so if you don't include the header row Excel will show the first unique value twice.

c. Extract Top Five (or nth) records

1. This method involves using a formula. When that is case, the criteria area does not have a header. If it does have a header, it should not match any of the fields in the table.
2. In the example, the formula is =F3>=LARGE(\$F\$3:\$F\$274,5) and is placed in I3. It will resolve to FALSE.
 - (1) For the bottom five: =F3>=SMALL(\$F\$3:\$F\$274,5)
3. Otherwise, you follow the same procedure you would as though there were a header. For the criteria, you will highlight the blank cell and the formula. Then continue as normal

d. Other Formulas

1. To find numbers within numbers
 - (1) =ISNUMBER(FIND("21",B3))
2. Filter for unmatched data in two columns
 - (1) =C3<>D3
3. Filter for blanks in the data
 - (1) =C2=""

8) SUMPRODUCT

- a. SUMPRODUCT is the function that was used before Microsoft came up with COUNTIFS and SUMIFS. It allowed for multiple conditions based on the Array concept but without the array entry requirement. It is different from other Functions in that the ranges are separated either by an asterisk or a plus instead of a comma. Normal algebraic signs apply to the ranges. For example, a Range could be equal to, less than, greater than, etc. to a value or other field reference.
- b. Syntax: `=SUMPRODUCT((Range1=Criteria1)*(Range2=Criteria2)*...*RangeToSum)`.
 Notice that the RangeToSum is not in its own parenthesis. This tells the function that this is the Sum range.
 - ✓ If the syntax were `=SUMPRODUCT((Range1=Criteria1)*(Range2=Criteria2)*...)` without a range to sum then the function will return a count instead of a sum.
 - ✓ * means AND
 - ✓ + means OR
 - ✓ Note: Range sizes must be the same for each range. If you are using a prompt field instead of a field within the range then the formats must be the same
- c. Example:
`=SUMPRODUCT((dept="OSC15910")*(fund="11000")*(acct="46200")*amount)`
 - ✓ The columns (dept, fund, acct, amount) are named ranges.
 - ✓ The algebraic symbol is a qualifier (e.g., =, >, <, >=, <=, <>)
 - ✓ The criteria (in double quotes) can be entered as the content of the field ("OSC15910"), a field within the table (B5), or a field range located outside of the table (This method is preferred since it serves to document your conditions for an outsider reviewer to see.)

9) Conditional Formatting – Dynamic Row Banding

- a. In the Intermediate class, students were shown how to place a line divider separating like items. In this exercise, we will create a dropdown list that will be used to determine which rows should be highlighted.
 1. Create a unique list. This was shown in the "Filter" exercise. This exercise has provided one unique lists in columns Q. You can create more.
 2. Create a dropdown menu.
 - (1) Place you cursor in cell B4
 - (2) Navigation: Data > Data Validation > Data Validation (Data Validation dialog opens) Settings (tab) > List (Allow dropdown)
 - (3) Click the range icon to the right of the Source field.
 - (4) Highlight the data in column Q ~ Click the range icon to the right.
 - (5) Click OK
 3. Create the conditional Format
 - (1) Highlight the range: A9:O130
 - (2) Navigation: Format > Conditional Formatting (dialog opens) > Formula is (from dropdown)
 - (3) SID: =\$C9=\$B\$4
The important thing to remember is that in the data table you need only lock the column reference (\$C9), but in the criteria area you must lock both column and row (\$B\$4)
 - (4) Click the Format Icon to the right: select the Patterns tab.
√ Select a shade
 - (5) Click OK twice and test your work

10) DataBase Functions

- a. The DataBase function was developed by Lotus for their spreadsheet program. Excel lists twelve database functions in its library
 1. DAVERAGE ~ Average
 2. DCOUNT ~ Counts Numbers
 3. DCOUNTA ~Counts cell entries
 4. DGET ~ Extracts a single record based on criteria
 5. DMAX ~ Largest value
 6. DMIN ~ Smallest value
 7. DPRODUCT ~ Multiplies values in the Field column
 8. DSTDEV ~ Standard Deviation
 9. DSTDEVP ~ Standard Deviation of an entire population
 10. DSUM ~ Sums based on criteria
 11. DVAR ~ Variance
 12. DVARP ~ Variance of an entire population
- b. The syntax for all database functions is: **=DSUM(DataSource,FieldofInterest,Criteria)**
 1. DataSource – The Data Source is the entire list of information you need to examine, including the field names at the top of the columns. This can be a defined table or a highlighted range.
 2. FieldofInterest – The Field of Interest is where the answer located. It can be entered on the formula as the column header (“Amount”), column number (4), or cell reference (D10).
 3. Criteria - The Criteria is made up of two types of information and is a separate area similar to the advanced filter. The first set of information is the name, or names, of the Fields(s) to be used as the basis for selecting the records. In this example it is the column header. The second set of information is the actual field value(s) or record(s), which are to be selected.
 - (1) All database functions have the same syntax: Data Source, Field of Interest, Criteria. It is important to understand that while you do not need to duplicate the column headings as was done for this example, copying the header row insures that the spelling is correct. Correct spelling is critical.
 4. This example uses two rows below the Criteria header for the actual criteria. The syntax is: Row1 AND criteria 1 AND criteria 2 AND ... Criteria N **OR** Row2 AND criteria 1 AND criteria 2 AND ... Criteria N
 - (1) To include multiple AND criteria for one column header, add that column header again to the right of the criteria: place your second AND criteria below: and make sure that extra column is included in the Criteria area.

5. Examples: Each of the following use different Criteria setups, but all yield the same answer.
- (1) =DSUM(dTbl,D10,A6:K8) (uses the cell reference for the criteria)
 - (2) =DSUM(dTbl,4,A6:K8) (uses the column reference for the criteria)
 - (3) =DSUM(dTbl,"Amount",A6:K8) (uses the column Name for the criteria)

11) Miscellaneous Items

a. Remove Duplicates

1. This step can be done using the Advanced Filter, but this is quicker. Copy the column with the data that you want to filter. Paste to a separate location.
2. Navigate: Data (Ribbon) > Remove Duplicates (Data Tools). The Remove Duplicates dialog displays. This will identify the data you have highlighted. Check the My Data Has Headers checkbox if applicable. Click OK.

b. Using TEXT

1. Syntax: =TEXT(Value,Format_Text)
 - (1) Value can be a cell reference (e.g., A1) or a number.
 - (2) Format_Text is how you would like the number displayed. The format always starts and ends with a double quote (“”) and may consist of any combination of the following
 - (a) d, dd, ddd, dddd
 - (i) d returns a serial number showing the day of the month.
 - (ii) dd returns a two digit serial number showing the day of the month.
 - (iii) ddd returns the day of the week as an abbreviation (Mon, Tue)
 - (iv) dddd returns the day of the week as a word (Monday)
 - (b) m, mm, mmm, mmmm, mmmmm
 - (i) m returns a serial number showing the month (1 = January)
 - (ii) mm returns a two digit serial number showing the month (01 = January)
 - (iii) mmm returns the month as an abbreviation (Jan, Feb)
 - (iv) mmmm returns the month as a word (January)
 - (v) mmmmm returns the first letter of the month (J)
 - (c) yy, yyyy
 - (i) yy returns the year as two digits
 - (ii) yyyy returns the year as four digits
 - (3) Examples
 - (a) =TEXT(41000,"m/dd/yyyy") returns 4/01/2012
 - (b) =TEXT(A1,,"m/dd/yyyy") where A1 = 41000 would also return 4/01/2012
 - (c) =TEXT(A1,"dddd, mm/dd/yyyy") returns Sunday, 04/01/2012

(d) ="Well this is a fine how do you do on "&TEXT(A1,"dddd, mm/dd/yyyy")
returns, Well this is a fine how do you do on Sunday, 04/01/2012

c. Copy As Values

1. Using TEXT assumes the use of a helper column where the source information is a numerical format and the answer is a text format. If you only want the text values and can't have two columns then you must the convert the formula to a value.
 - (1) Highlight the cells you want to convert to values.
 - (2) Mouse over the right border of the highlighted data.
 - (3) When your cursor changes to a four direction arrow, then right click and drag the highlighted area to the right on cell's distance and then back again, letting go of the mouse to drop the highlighted area on top of itself.
 - (a) A menu will display and the third item will be Copy Here as Values only. Select this option.
 - (4) Delete the source cells.

12) FRP301 Formula

a. 01-Intro

1. =COUNTIF(Range,Criteria)
2. =SUMIF(Range,Criteria,RangeToSum)
3. =COUNTIFS(CriteriaRange1,Criteria1,CriteriaRange2,Criteria2) Excel 2007 and later
4. =COUNTIFS(TRC,\$J24,DAY,K\$23) ~ example from the exercise
5. =SUMIFS(RangeToSum,CriteriaRange1,Criteria1,CriteriaRange2,Criteria2) Excel 2007 and later
 - (1) =SUMIFS(Quantity,TRC,\$J37,DAY,K\$36) ~ example from the exercise

b. 02-IF

1. =IF(ConditionToTest,IfTrue,IfFalse)
2. =IF(AND(Condition1,Condition2, ...,Condition30),"IfTrue,IfFalse)
3. =IF(OR(Condition1,Condition2, ...,Condition30),"IfTrue,IfFalse)
4. =IF(ISERROR(FormulaToBeEvaluated),"AnswerIfTrue",AnswerIfFalse)
 - (1) (example) =IF(ISERROR(H3/D3)," ",H3/D3)
5. =IFERROR(FormulaToBeEvaluated," AnswerIfError") Excel 2007/2010
 - (1) (example) =IFERROR(H3/D3,"")
6. Multi-condition
 - (1) =IF(Condition,IfTrue,[IfFalse],IF(Condition,IfTrue,[IfFalse],IF(Condition,IfTrue,IfFalse,)))
7. Multi-condition using VLOOKUP
 - (1) =IF(ISERROR(O3*VLOOKUP(E3,percent,2,0)),"",O3*VLOOKUP(E3,percent,2,0))
 - (2) =IFERROR(O3*VLOOKUP(E3,table,2,0),"") Excel 2007/2010

c. 03-VLOOKUP

1. =VLOOKUP(ItemToFind,RangeToLookIn,ColumnToPickFrom,Unsorted[false]OrSorted[true])

2. Examples from class
 - (1) =VLOOKUP(B19,character,2,FALSE)
 - (2) =VLOOKUP(I19,grades,2,TRUE)
 - (3) =VLOOKUP(\$B4,CT_tbl,C\$2,TRUE)
 - (4) =VLOOKUP(A5+0,tbl_1,2,FALSE)
 - (a) =VLOOKUP(A5*1,tbl_1,2,FALSE)
 - (b) =VLOOKUP(--A5,tbl_1,2,FALSE)
 - (c) =VLOOKUP(TEXT(H5,"0"),TBL_2,2,FALSE)

d. Formulae used to determine Fiscal periods

1. State Fiscal Period
 - (1) =IF(MONTH(B4)+6>12,MONTH(B4)-6,MONTH(B4)+6) or
 - (2) =CHOOSE(TEXT(B4,"m"),7,8,9,10,11,12,1,2,3,4,5,6)
2. Federal Fiscal Period
 - (1) =IF(MONTH(B4)+3>12,MONTH(B4)-9,MONTH(B4)+3) or
 - (2) =CHOOSE(TEXT(H4,"m"),4,5,6,7,8,9,10,11,12,1,2,3)
3. State Fiscal Year
 - (1) =IF(MONTH(B4)+6>12,YEAR(B4)+1,YEAR(B4))
4. Federal Fiscal Year
 - (1) =IF(MONTH(B4)+3>12,YEAR(B4)+1,YEAR(B4))
5. State Fiscal Quarter
 - (1) =MOD(CEILING(22-7+MONTH(B4),3)/3,4)+1 or
 - (2) =CHOOSE(TEXT(B4,"m"),3,3,3,4,4,4,1,1,1,2,2,2)
6. Federal Fiscal Quarter
 - (1) =MOD(CEILING(22-10+MONTH(B4),3)/3,4)+1 or
 - (2) =CHOOSE(TEXT(H4,"m"),2,2,2,3,3,3,4,4,4,1,1,1)
7. Substitute your target cell for B4 in each example

e. Miscellaneous Formula

1. =ROW(A1) equals 1
2. =ROW(A2) equals 2 etc.
3. =COLUMN(A1) equals 1
4. =COLUMN(B1) equals 2 etc.